


Online Appendix
Simple Model for Welfare impacts of CTBAs vs Carbon taxes
$ontext
This model has one commodity plus a numeraire, to assess the welfare implications of using either a production or a consumption tax in one country on global welfare.  A key question is whether the welfare impacts are best described by the elasticity of demand/supply, or the combined elasticities of supply and demand.
$offtext
SETS
*Home and foreign ctries.  Emission and clean.  Numeraire and non
Ctry   /H,F/
Commod /E,C/
;
PARAMETERS
PN          Price of numeraire  /1 /
Beta1       Base supply share of commodity /0.5 /
Alpha       Base demand share of non-numeraire commodity /0.5 /
sigmat      Elasticity of transformation in production/ 0.9 / sigma       Elasticity of substitution in demand /0.9 /
Q           Country resource endowments   /100/
TC(Ctry)    Consumption tax rate  /H 0, F 0/
TP(Ctry)    Production tax rate   /H 0, F 0/
U0(Ctry)    Base utility level    /H 100,F 100/
;
VARIABLES
PI          World price of good 1
SD(Ctry)    Supply of the good from each ctry
SDT         Global Supply and emissions
DD(Ctry)    Demand for good by each ctry
U(Ctry)     Utility level
EXP(Ctry)   Expenditure function
CPI(Ctry)   Consumer price index
Rev(Ctry)   revenue from production in each country
GREV(Ctry)  Govt revenue by country
GLOBEV      EV globally
;
*Initializing Variables
PI.L = 1;
U.L(Ctry)=100; CPI.L(Ctry)=1;
EQUATIONS

 SUPPLY(Ctry)
 SDTOT
 DEMAND(Ctry)
 SD_Balance
 EXPDEF(Ctry)
 CPIDEF(ctry)
 REVDEF(Ctry)
 GREVDEF(ctry)
 UDef(ctry)
 GLOBEVDef
;
SUPPLY(Ctry)..  SD(Ctry) =e= beta1*(PI*(1+TP(ctry))/PN)**(sigmat)*Q ;
SDTOT..      SDT =e= Sum(Ctry,SD(Ctry)) ;
DEMAND(Ctry).. DD(Ctry) =e= alpha*(PI*(1+TC(ctry))/PN)**(-Sigma)*U(ctry) ;
SD_Balance..   Sum(ctry,SD(ctry)) =e= Sum(ctry,DD(ctry)) ;
EXPDEF(Ctry)..
EXP(ctry) =e=  (alpha*(PI*(1+TC(ctry)))**(1-sigma) + (1-alpha)*PN**(1-sigma))**(1/(1-sigma))*U(ctry);
CPIDEF(ctry)..
CPI(Ctry) =e= (alpha*(PI*(1+TC(ctry)))**(1-sigma) + (1-alpha)*PN**(1-sigma))**(1/(1-sigma)) ;
REVDEF(Ctry)..
   Rev(ctry) =e=   (Beta1*(PI*(1+TP(ctry)))**(1+sigmat) + (1-Beta1)*PN**(1+sigmat))**(1/ (1+sigmat))*Q;
GREVDEF(ctry)..
GREV(Ctry)  =e= TC(ctry)*PI*DD(Ctry) - TP(ctry)*PI*SD(Ctry);
UDef(ctry)..   U(ctry) =e= (REV(ctry) + GREV(ctry))/CPI(ctry) ;
GLOBEVDef..    GLOBEV =e= Sum(ctry,u(ctry)) ;
*Model
Model Effic_Cost     efficiency costs of consumer and Producer tax/ all  /;
Options ITERLIM = 1000, LIMROW = 50, LIMCOL = 50, SOLPRINT=ON, MCP=PATH, 
NLP=CONOPT4, CNS= Conopt4
;
 SOLVE Effic_Cost Using CNS;
*Now solving with a tax on demand
 TC(“H”) = 0.2;
 SOLVE Effic_Cost Using CNS;
*Now with a Prodn tax-- last results are pdn tax
 TC(“H”) = 0.0;
 TP(“H”) = -0.2;
  SOLVE Effic_Cost Using CNS;

